

The objective of Lab 1 is to introduce you to the SAS statistical analysis program, to demonstrate how to enter data into SAS, and to show you how to perform basic descriptive statistical analysis of data from two groups.

1. Introduction to SAS (**SASstart.pdf**): data is stored in **tomato.txt** and SAS code is **tomato.sas**.
2. Creative Writing Example – This data set we will study comes from Section 1.1 of *The Statistical Sleuth*. In this study, the researcher wanted to explore the connection between motivation and creativity. **Do experienced creative writers show different levels of creativity based on the source of motivation for their writing, either intrinsic or extrinsic?** To answer this research question, a group of experienced creative writers was randomly assigned to one of two treatments. Both treatments involved giving the writers a questionnaire on motivation for writing. For one group of writers, the questionnaire asks about the intrinsic motivation for writing (enjoyment, satisfaction, etc.). For the other group of writers, the questionnaire asks about the extrinsic motivation for writing (jobs, financial rewards, awards, etc.). After completing the survey, all subjects were assigned to write a Haiku style poem on laughter. A group of 12 poets then scored these poems on a 40-point scale based on the creativity shown in the writing.

cwrite.sas

- a. Enter the data into SAS – Data entry into SAS can be accomplished through the data step command in several ways. The file **cwrite.sas** shows an example of manually entering the data set. In this example, the data are saved to the name **set1** and consist of three variables: the subject number (**subject** = 1 – 47), the treatment received (**trt** = 1 for intrinsic or 2 for extrinsic), and the value of the response variable (**y** = score received on creativity displayed in writing a Haiku style poem on laughter).

Below is a section of the data step command:

```
data set1;
input subject trt y;
datalines;
1 1 12.0
2 1 12.0
3 1 12.9
4 1 13.6
5 1 16.6
: : :
46 2 22.1
47 2 24.0
run;
```

It also shows an example of reading in the data set from the **cwrite.csv** file. Again, the data are saved to the name **set1**.

```
data set1;
infile 'cwrite.csv' dlm=',' firstobs=2;
input subject trt y;
run;
```

- b. More data entry – In this example, specifying the variable **trt** as either the value 1 or 2 is not very descriptive. We can use **proc format** to specify what the values 1 and 2 represent. The command is:

```
proc format;  
value trt 1='intrinsic'  
          2='extrinsic';  
run;
```

Finally, we can sort the data set by the value of any variable using **proc sort**. The command to sort the data set **set1** by the variable **trt** is:

```
proc sort data=set1; by trt;  
run;
```

- c. Printing the data set to the output window – If you want to include the data set in your output, you can use **proc print** as:

```
proc print data=set1; run;
```

- d. Obtaining Summary statistics – Summary statistics for the response variable for each treatment group can be obtained using **proc univariate**. Responses for the two treatment groups are split using the **by** command.

```
proc univariate data=set1; by trt;  
  var y;  
run;
```

- e. Boxplots – While summary statistics are nice to use to compare values of the response variable between the two groups, a visual representation of these values would also be helpful. There are several options to choose from, but side-by-side boxplots are one of the best in determining differences in values between groups. The command below also includes formatting commands to label the axes.

```
proc sgplot data=set1;  
hbox y / category=trt datalabel=subject nofill;  
format trt trt.;  
xaxis label="Creativity Evaluation" valueattrs=(size=15)  
      labelattrs=(size=20);  
yaxis label="Treatment" valueattrs=(size=15)  
      labelattrs=(size=20);  
run;
```

- f. Running the analysis – To run the analysis and obtain all output, we will need to submit the commands to SAS. To do this, click on the “running person” button towards the top of the window. SAS will run the commands in the file **cwrite.sas** and show the results of the analysis in the results window.